
	Reason for change:
	In the current TTCN, at Step 6 the PDU session Release command with cause #26 is transmitted, and the special messge content also sets the T3396 to 0. After the UE transimts the PDU Session release Complete message at Step 7, the SS releases the rrc connection.

As per 24.501 cl 6.3.3.3

Quote

If the PDU SESSION RELEASE COMMAND message includes 5GSM cause #26 "insufficient resources" and the Back-off timer value IE, the UE shall ignore the 5GSM congestion re-attempt indicator IE provided by the network, if any, and the UE shall take different actions depending on the timer value received for timer T3396 in the Back-off timer value:
<<skipped text>>

c)
if the timer value indicates zero, the UE:
1)
shall stop timer T3396 associated with the corresponding DNN, if running, and may send a PDU SESSION ESTABLISHMENT REQUEST message or PDU SESSION MODIFICATION REQUEST message for the same DNN; and
2)
if no DNN was provided during the PDU session establishment and the request type was different from "initial emergency request" and different from "existing emergency PDU session", the UE shall stop timer T3396 associated with no DNN, if running, and may send a PDU SESSION ESTABLISHMENT REQUEST message without a DNN, or a PDU SESSION MODIFICATION REQUEST message without an DNN provided by the UE.
Unquote

Thus, it might happen that before the user gets a chance to trigger a PDU session according to Step8, the UE may transmit a rrcREQ for establishing a PDU session. This is not handled in current TTCN implementation.

Also, the TP2 is achieved by transmitting a PDU Session Release Command with cause #29 at Step 18, AFTER a successful PDU Session Authentication Command/Complete is performed at Steps 16 and 17.

It can be seen that the same procedure (PDU Session Authentication Command/Complete) is also performed at Steps 4 -5. The steps (6 – 17) after this don’t really serve any purpose for TP testing and are only adding additional signalling that is avoidable. Moreover, voiding the steps 6-17 will also avoid the issue highlighted above with cause #26 (Steps 6-17). A Prose CR will be raised for this.

At Step 18, the PDU session release command should also contain EAP-failure IE. This is in-line with TP2 description. The special message content for Step 18 should include this. Current TTCN implementation does not include this IE.

According to the preamble only one PDU session is established in the testcase which is being played with. Some UEs may try to initiate a PDU session after Step 19, as there will be no PDU session active. Current TTCN does not handle such a situation.

To avoid such a possibility, it is proposed that the preamble be changed to state 3N-A, and New steps 0A and 0B are introduced where a second PDU session is established to be played with.

A Prose CR will be raised for this.

	
	

	Summary of change:
	Re-worked the whole testcase implementation for 10.1.1.2

	A conformant UE will fail the testcase.

1 Corrections required
1.1 function fl_TC_10_1_1_2_Body()
	Function name
	fl_TC_10_1_1_2_Body

	Reason for change
	In the current TTCN, at Step 6 the PDU session Release command with cause #26 is transmitted, and the special messge content also sets the T3396 to 0. After the UE transimts the PDU Session release Complete message at Step 7, the SS releases the rrc connection.

As per 24.501 cl 6.3.3.3

Quote

If the PDU SESSION RELEASE COMMAND message includes 5GSM cause #26 "insufficient resources" and the Back-off timer value IE, the UE shall ignore the 5GSM congestion re-attempt indicator IE provided by the network, if any, and the UE shall take different actions depending on the timer value received for timer T3396 in the Back-off timer value:
<<skipped text>>

c)
if the timer value indicates zero, the UE:
1)
shall stop timer T3396 associated with the corresponding DNN, if running, and may send a PDU SESSION ESTABLISHMENT REQUEST message or PDU SESSION MODIFICATION REQUEST message for the same DNN; and
2)
if no DNN was provided during the PDU session establishment and the request type was different from "initial emergency request" and different from "existing emergency PDU session", the UE shall stop timer T3396 associated with no DNN, if running, and may send a PDU SESSION ESTABLISHMENT REQUEST message without a DNN, or a PDU SESSION MODIFICATION REQUEST message without an DNN provided by the UE.
Unquote

Thus, it might happen that before the user gets a chance to trigger a PDU session according to Step8, the UE may transmit a rrcREQ for establishing a PDU session. This is not handled in current TTCN implementation.

Also, the TP2 is achieved by transmitting a PDU Session Release Command with cause #29 at Step 18, AFTER a successful PDU Session Authentication Command/Complete is performed at Steps 16 and 17.

It can be seen that the same procedure (PDU Session Authentication Command/Complete) is also performed at Steps 4 -5. The steps (6 – 17) after this don’t really serve any purpose for TP testing and are only adding additional signalling that is avoidable. Moreover, voiding the steps 6-17 will also avoid the issue highlighted above with cause #26 (Steps 6-17). A Prose CR will be raised for this.

At Step 18, the PDU session release command should also contain EAP-failure IE. This is in-line with TP2 description. The special message content for Step 18 should include this. Current TTCN implementation does not include this IE.



	Summary of change
	Re-worked the implementation for fl_TC_10_1_1_2_Body

	TTCN module
	NR5GC\10_1\GSM_TestCases_NR5GC.ttcn

	MCC160 Comment
	


Before Change:

	function fl_TC_10_1_1_2_Body() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA));
    var template (value) GPRS_Timer3 v_BackOffTimer_Step6_18 := cs_GPRS_Timer2_3_IEI('5E'O, '101'B, '00000'B);
    var PDUSessionInfoList_Type v_PDUSessionInfo;
    var NG_PDU_SessionId v_PDUSessionIdToRelease;
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (v_NGC_NASCellA); //WA#R5s200401
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams); //WA#R5s200401
    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_RadioBearerId_Type v_SRB;
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var charstring v_NSSAI:= f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI_SST1eMBB);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;    
    v_GSM_MobilityInfo := v_PDUSessionInfoList[0];    
    v_SRB := tsc_NR_RbId_SRB2;
    //@siclog "Steps 1 - 2" siclog@
    //Send PDU Session Authentication Command and receive PDU Session Authentication Complete
    f_PDUSessionAuthentication( v_NGC_NASCellA, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    //@siclog "Step 3" siclog@ @sic R5-200214 sic@
    //SS transmits PDU Session Authentication result
    v_MsgToSend := f_Get_PDUSessionAuthenticationResult(v_GSM_MobilityInfo, cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success))));
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    //@siclog "Steps 4 - 5" siclog@
    //Send PDU Session Authentication Command and receive PDU Session Authentication Complete
    f_PDUSessionAuthentication( v_NGC_NASCellA, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_PDUSessionIdToRelease := v_PDUSessionInfo[f_NR5GC_MobileInfo_GetNoOfSessions()-1].SessionId;
    //@siclog "Step 6" siclog@
    // send PDU Session Release Command with cause "insufficient resources"
    v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease, -, v_BackOffTimer_Step6_18);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    //@siclog "Steps 7" siclog@
    // The UE transmits a PDU SESSION RELEASE Complete
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {
    }
    //@siclog "Step 7A" siclog@
    // SS releases the rrc connection
    f_NR_RRCRelease(v_NGC_NASCellA);
    // Need to 'reset' all stored lists as all DRBs and PDU sessions no longer exist in UE
    vc_NR5GC_Global.MobileInfo.ListOfDRBs := {}; //WA#R5s200401
    //@siclog "Steps 8" siclog@
    //Cause the UE to request establishment of PDU session.    
    f_UT_RequestPDUSession(UT, 3, tsc_RandomAPN, -, -, v_NSSAI);    
    //@siclog "Steps 9 - 12" siclog@
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?); //WA#R5s200401
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu,  cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), //WA#R5s200401
                                      tsc_NasKsi_NativeSecurityContext),
                                      v_S_TMSI, '0000'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR_RRC_ActivateSecurity (v_NGC_NASCellA, valueof(v_ReceivedMsg.SecurityProtection.NasCount)); // Use NasCount returned from Service Req //WA#R5s200401
    f_NR_SendRRCReconfiguration(v_NGC_NASCellA, -, cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB2)}, -, -,  //WA#R5s200401
                                cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master)),
                                cs_NR_RRCReconfiguration_v1530_IEs(omit),
                                {cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_SERVICE_ACCEPT(v_ReceivedMsg.Pdu.Msg.service_Request.pduSessionStatus))});
    //@siclog "Steps 13 - 15a1" siclog@
    f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1);//WA#R5s200697
    v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_GSM_MobilityInfo := v_PDUSessionInfoList[0];
    //@siclog "Steps 16 - 17" siclog@
    //Send PDU Session Authentication Command and receive PDU Session Authentication Complete
    f_PDUSessionAuthentication( v_NGC_NASCellA, f_NR5GC_Security_Get(), v_GSM_MobilityInfo, v_PLMN);
    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_PDUSessionIdToRelease := v_PDUSessionInfo[f_NR5GC_MobileInfo_GetNoOfSessions()-1].SessionId;
    //@siclog "Steps 18" siclog@
    // send PDU Session Release Command with cause #29 user authentication or authorization failed
    v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease, cs_GMM_GSM_Cause (omit, '00011101'B), v_BackOffTimer_Step6_18);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    //@siclog "Steps 19" siclog@
    // The UE transmits a PDU SESSION RELEASE Complete
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 19");
    } //End of test body of TC_10_1_1_2


After Change:

	function fl_TC_10_1_1_2_Body() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA));
    var template (value) GPRS_Timer3 v_BackOffTimer_Step6_18 := cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00000'B);
    var PDUSessionInfoList_Type v_PDUSessionInfo;
    var NG_PDU_SessionId v_PDUSessionIdToRelease;    
    var NG_NAS_SecurityParams_Type v_SecurityParams := f_NR5GC_Security_Get();
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_RadioBearerId_Type v_SRB;
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    v_SRB := tsc_NR_RbId_SRB2;    
    //@siclog "Steps 0A - 0B" siclog@
    f_UT_RequestPDUSession(UT, 3, tsc_RandomAPN); //WA#10_1_1_2    
    f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1); //WA#10_1_1_2
    v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();  //WA#10_1_1_2
    v_GSM_MobilityInfo := v_PDUSessionInfoList[1]; //WA#10_1_1_2
    //@siclog "Steps 1 - 2" siclog@
    //Send PDU Session Authentication Command and receive PDU Session Authentication Complete
    f_PDUSessionAuthentication( v_NGC_NASCellA, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN, SRB1_2 /*WA#10_1_1_2*/);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    //@siclog "Step 3" siclog@ @sic R5-200214 sic@
    //SS transmits PDU Session Authentication result
    v_MsgToSend := f_Get_PDUSessionAuthenticationResult(v_GSM_MobilityInfo, cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success))));
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    //@siclog "Steps 4 - 5" siclog@
    //Send PDU Session Authentication Command and receive PDU Session Authentication Complete
    f_PDUSessionAuthentication( v_NGC_NASCellA, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN, SRB1_2 /*WA#10_1_1_2*/);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    v_PDUSessionInfo := f_NR5GC_MobileInfo_GetSessionInfoList();
    v_PDUSessionIdToRelease := v_PDUSessionInfo[f_NR5GC_MobileInfo_GetNoOfSessions()-1].SessionId;
    //@siclog "Step 6 - 17" siclog@

//VOIDED : A Prose CR will be raised.    
    //@siclog "Steps 18" siclog@
    // send PDU Session Release Command with cause #29 user authentication or authorization failed
    v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease, 
    











 cs_GMM_GSM_Cause (omit, '00011101'B), 
    











 v_BackOffTimer_Step6_18,

 cs_EAP_Message(bit2oct(encvalue(cs_EAP_FAilure))));
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    //@siclog "Steps 19" siclog@
    // The UE transmits a PDU SESSION RELEASE Complete
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
    if (not f_Check_NG_PDUSessionReleaseComplete (v_PDUSessionInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 19");
    } //End of test body of TC_10_1_1_2


1.2 function f_TC_10_1_1_2_NR5GC()
	Function name
	f_TC_10_1_1_2_NR5GC

	Reason for change
	According to the preamble only one PDU session is established in the testcase which is being played with. Some UEs may try to initiate a PDU session after Step 19, as there will be no PDU session active. Current TTCN does not handle such a situation.

To avoid such a possibility, it is proposed that the preamble be changed to state 3N-A, and New steps 0A and 0B are introduced where a second PDU session is established to be played with.

A Prose CR will be raised for this.

	Summary of change
	Instead of calling function f_NR5GC_Preamble (v_NGC_NASCellA, STATE_IDLE_1A, -, -, noRrcConnectionRelease);, called function f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A); for preamble.

	TTCN module
	NR5GC\10_1\GSM_TestCases_NR5GC.ttcn

	MCC160 Comment
	


Before Change:

	function f_TC_10_1_1_2_NR5GC() runs on NR5GC_PTC
  {
    //UE-requested PDU session release / Abnormal / Collision with network-requested PDU session modification procedure
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    f_NR5GC_Init_NAS(NR_1);
    //Create and configure cells
    f_NR_CellConfig_Def(v_NGC_NASCellA);
    //Preamble
    f_NR5GC_Preamble (v_NGC_NASCellA, STATE_IDLE_1A, -, -, noRrcConnectionRelease);
    //Main test body
    fl_TC_10_1_1_2_Body();
    // Finish
    f_NR_Postamble(v_NGC_NASCellA, STATE_CONNECTED_3A);


After Change:

	function f_TC_10_1_1_2_NR5GC() runs on NR5GC_PTC
  {
    //UE-requested PDU session release / Abnormal / Collision with network-requested PDU session modification procedure
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    f_NR5GC_Init_NAS(NR_1);
    //Create and configure cells
    f_NR_CellConfig_Def(v_NGC_NASCellA);
    //Preamble
    //WA#10_1_1_2 f_NR5GC_Preamble (v_NGC_NASCellA, STATE_IDLE_1A, -, -, noRrcConnectionRelease);
    f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A); //WA#10_1_1_2
    //Main test body
    fl_TC_10_1_1_2_Body();
    // Finish
    f_NR_Postamble(v_NGC_NASCellA, STATE_CONNECTED_3A);
  }


1.3 function f_PDUSessionAuthentication
	Function name
	f_PDUSessionAuthentication

	Reason for change
	This function is called for Steps 1-2 and 4-5 respectively. This is called after the SRB2 has been established (This happens when PDU session is established).
However, in this function, the PDU Session Authentication command is sent on SRB1 instead of SRB2. The PDU Session Authentication Complete is also expected on SRB1 instead of SRB2. 

This needs to be corrected.

	Summary of change
	Added a new argument to the function and based on that it is decided whether the SRB1 is to be used or not.
Please see below.

	TTCN module
	NR5GC\Common\NR5GC\NR5GC_NASSteps.ttcn

	MCC160 Comment
	


Before Change:

	function f_PDUSessionAuthentication(NR_CellId_Type p_CellId,
                                      NG_NAS_SecurityParams_Type p_SecurityParams,
                                      GSM_MobilityInfo_Type p_GSM_MobilityInfo,
                                      NAS_PlmnId     p_PLMN)
    runs on NR_BASE_PTC return NG_NAS_SecurityParams_Type
  {
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_SecurityParams_Type v_SecurityParams := p_SecurityParams;
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;
    // send AUTHENTICATION COMMAND
    v_MsgToSend := f_Get_PDUSessionAuthenticationCmd(v_SecurityParams, p_GSM_MobilityInfo, p_PLMN);
    SRB.send(cas_NR_SRB_NasPdu_REQ(p_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    // receive AUTHENTICATION COMPLETE
    SRB.receive(car_NR_SRB_NasPdu_IND(p_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
    // To be completed
    if (not f_Check_NG_PDUSessionAuthenticationComplete(p_GSM_MobilityInfo, v_SecurityParams, v_ReceivedAsp.Signalling.Nas[0].Pdu, *, *)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "GSM Authentication Failed");
    }
    return v_SecurityParams;
  }


After Change:

	function f_PDUSessionAuthentication(NR_CellId_Type p_CellId,
                                      NG_NAS_SecurityParams_Type p_SecurityParams,
                                      GSM_MobilityInfo_Type p_GSM_MobilityInfo,
                                      NAS_PlmnId     p_PLMN,
                                      SRBs_AlreadyConfigured p_SRB := SRB1_2/*WA#10_1_1_2*/)
    runs on NR_BASE_PTC return NG_NAS_SecurityParams_Type
  {
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_SecurityParams_Type v_SecurityParams := p_SecurityParams;
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1; 
    if (p_SRB == SRB1_2) {v_SRB := tsc_NR_RbId_SRB2;} //WA#10_1_1_2
    // send AUTHENTICATION COMMAND
    v_MsgToSend := f_Get_PDUSessionAuthenticationCmd(v_SecurityParams, p_GSM_MobilityInfo, p_PLMN);
    SRB.send(cas_NR_SRB_NasPdu_REQ(p_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    // receive AUTHENTICATION COMPLETE
    SRB.receive(car_NR_SRB_NasPdu_IND(p_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;
    // To be completed
    if (not f_Check_NG_PDUSessionAuthenticationComplete(p_GSM_MobilityInfo, v_SecurityParams, v_ReceivedAsp.Signalling.Nas[0].Pdu, *, *)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "GSM Authentication Failed");
    }
    return v_SecurityParams;
  }


1.4 function f_NR5GC_PDUSessionEstablishment_InnerLoop
	Function name
	f_NR5GC_PDUSessionEstablishment_InnerLoop

	Reason for change
	The function f_PDUSessionAuthentication is also called in function f_NR5GC_PDUSessionEstablishment_InnerLoop, and this has been modified in change 2.2 above.

As a consequence, the function call to f_PDUSessionAuthentication in function f_NR5GC_PDUSessionEstablishment_InnerLoop also needs to be updated.

	Summary of change
	Updated the call to function f_PDUSessionAuthentication within function f_NR5GC_PDUSessionEstablishment_InnerLoop.
Please see below.

	TTCN module
	NR5GC\Common\NR5GC\NR5GC_NASSteps.ttcn

	MCC160 Comment
	


Before Change:

	<<SKIPPED CODE>>
if(p_PDUSessionAuthenticationRequired){ // @sic R5-198184 sic@
    //Perform PDU Session Authentication if required
    v_SecurityParams := f_NR5GC_Security_Get();
    f_PDUSessionAuthentication( p_CellId, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN);
    }
<<SKIPPED CODE>>


After Change:

	<<SKIPPED CODE>>

if(p_PDUSessionAuthenticationRequired){ // @sic R5-198184 sic@
    //Perform PDU Session Authentication if required
    v_SecurityParams := f_NR5GC_Security_Get();
    f_PDUSessionAuthentication( p_CellId, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN, p_SRB /*WA#10_1_1_2*/);
    }

<<SKIPPED CODE>>


